

KROY K4500

**THERMAL TRANSFER / DIRECT THERMAL
BAR CODE PRINTER**

**PROGRAMMING
MANUAL**

TABLE OF CONTENTS

DOCUMENT CONVENTIONS.....	I
SETUP AND SYSTEM COMMANDS.....	1
SIZE.....	1
GAP.....	2
BLINE.....	4
OFFSET.....	5
SPEED.....	6
DENSITY.....	7
DIRECTION.....	8
REFERENCE.....	9
COUNTRY.....	10
CODEPAGE.....	11
FEED.....	12
BACKFEED.....	13
FORMFEED.....	14
HOME.....	15
PRINT.....	16
CUT.....	18
REM.....	19
LABEL FORMATTING COMMANDS.....	20
CLS.....	20
BAR.....	21
ERASE.....	22
REVERSE.....	23
BOX.....	24
BITMAP.....	25
PUTBMP.....	26
PUTPCX.....	27
BARCODE.....	28
DMATRIX.....	32
MAXICODE.....	33
PDF417.....	35
TEXT.....	36
STATUS POLLING COMMANDS (RS-232).....	37
<ESC>!?.....	37
<ESC>!R.....	38
~!T.....	39
~!I.....	40
~!F.....	41
MESSAGE TRANSLATION PROTOCOLS.....	42
~#.....	42
INPUT.....	43
OUT.....	45
BEEP.....	46
COMMANDS FOR WINDOWS DRIVER.....	47
!B.....	47
!J.....	48
FILE MANAGEMENT COMMANDS.....	49
DOWNLOAD.....	49
EOP.....	51
FILES.....	52
KILL.....	53
RUN.....	54
DEVICE RECONFIGURATION COMMANDS.....	55

SET COUNTER	55
SET CUTTER	56
SET PEEL	58
SET TEAR	59
SET GAP	60
SET RIBBON	61
SET COM1	62
PRINTER GLOBAL VARIABLES	63
@LABEL	63
@YEAR	64
@MONTH	65
@DATE	66
@DAY	67
@HOUR	68
@MINUTE	69
@SECOND	70

Document Conventions

This manual uses the following typographic conventions.

Convention	Description
[expression list]	Items inside square brackets are optional, expression maximum length 2*1024 bytes;
<ESC>	ESCAPE (ASCII 27), control code of status polling command, which returns the printer status immediately, no matter the printer is ready or not.
~	(ASCII 126), control code of status polling command, which returns the printer status only when the printer is ready.
Space , Tab	(ASCII 32),(ASCII 9) characters will be ignored in the command line
“	(ASCII 34), beginning and ending of expression
CR,LF	(ASCII 13),(ASCII 10) or (ASCII 10) is placed at the end of command line.
NULL	(ASCII 0) is not supported in the expression, except the 2D bar code commands.
\nnn	(ASCII 92), nnn is a number. Available range is from 000 to 255 to express the character in decimal.
FF , EOF	(ASCII 12),(ASCII 26) can be used at the end of command line only.
Note: <i>200 DPI: 1 mm = 8 dots</i>	Arial font in bold and italic type is used for note.

Setup and System Commands

- **SIZE**

Description

This command defines the label width and length.

Syntax


- (1) English system (inch)
SIZE m, n
- (2) Metric system (mm)
SIZE m mm, n mm

<u>Parameter</u>	<u>Description</u>
m	Label width (inch or mm)
n	Label length (inch or mm)

Note: 200 DPI: 1 mm = 8 dots
300 DPI: 1mm = 12 dots
For metric system, there must be a space between parameter and mm.

Example

- (1) English system (inch)
SIZE 3.5, 3.00
- (2) Metric system (mm)
SIZE 100 mm, 100 mm



See Also

GAP, BLINE

● GAP

Description

Define the gap distance between two labels

Syntax

- (1) English system (inch)
GAP m, n
- (2) Metric system (mm)
GAP m mm, n mm

Parameter

m

n

Description

The gap distance between two labels

$0 \leq m \leq 1$ (inch), $0 \leq m \leq 25.4$ (mm)

The offset distance of the gap


$n \leq$ label length (inch or mm)

Note: For metric system, there must be a space between parameter and mm.

Example

Normal gap

- (1) English system (inch)
GAP 0.12, 0
- (2) Metric system (mm)
GAP 3 mm, 0




Special gap

(1) English system (inch)

GAP 0.30, 0.10

(2) Metric system (mm)

GAP 7.62 mm, 2.54 mm



See Also

SIZE, BLINE

● BLINE

Description

This command is used to set the height of the black line and the user-defined extra label feeding length each form feed takes.

Syntax


- (1) English system (inch)
BLINE m, n
- (2) Metric system (mm)
BLINE m mm, n mm

<u>Parameter</u>	<u>Description</u>
m	The height of black line either in inch or mm. $0.1 \leq m \leq 1$ (inch), $2.54 \leq m \leq 25.4$ (mm)
n	The extra label feeding length. $0 \leq n \leq$ label length

Note: For metric system, there must be a space between parameter and mm.

Example

- (1) English system (inch)
BLINE 0.20, 0.50
- (2) Metric system (mm)
BLINE 5.08 mm, 12.7 mm



See Also

SIZE, GAP

● OFFSET

Description

This command defines the selective, extra label feeding length each form feed takes, which, especially in peel-off mode and cutter mode, is used to adjust label stop position, so as for label to register at proper places for the intended purposes. The printer backtracks the extra feeding length before the next run of printing.

Syntax


- (1) English system (inch)
OFFSET m
- (2) Metric system (mm)
OFFSET m mm

<u>Parameter</u>	<u>Description</u>
m	The offset distance (inch or mm) $-1 \leq m \leq 1(\text{inch})$

CAUTION: Improperly offset value may cause paper jam.

Example

- (1) English system (inch)
OFFSET 0.5
- (2) Metric system (mm)
OFFSET 12.7 mm



See Also

SIZE, GAP, SET PEEL, SET CUTTER

- **SPEED**

Description

This command defines the print speed.

Syntax

SPEED n

Parameter

n

Description

printing speed in inch per second

n	2	3	4	5	6	8	10	12
K4500			x		x	x	x	

Example

SPEED 10

See Also

DENSITY

- **DENSITY**

Description

This command designates the level of darkness of printing.

Syntax

DENSITY n

Parameter

n

Description

0, specifies the lightest level

15, specifies the darkest level

Example

DENSITY 7

See Also

DENSITY

- **DIRECTION**

Description

This command defines the print direction.

Syntax

DIRECTION n

Parameter

n

Description

0 or 1. Please refer to the illustrations below:

(DIRECTION 0)



(DIRECTION 1)



Example

DIRECTION 0


See Also

REFERENCE

● REFERENCE

Description

This command defines the reference point of the label. The reference (origin) point varies with the print direction, as shown:



Syntax

REFERENCE x, y

Parameter

x
y

Description

Horizontal coordinate, with “dot” as the unit.
Vertical coordinate, with “dot” as the unit.

Note: 200 DPI: 1 mm = 8 dots
300 DPI: 1 mm = 12 dots

Example

REFERENCE 10,10

See Also

DIRECTION

● COUNTRY

Description

This command defines what special character to have on the portable LCD keyboard (option) to orient the keyboard for use in different countries.

Syntax

COUNTRY n

Parameter

n

Description

001: USA
002: Canadian-French
003: Spanish (Latin America)
031: Dutch
032: Belgian
033: French (France)
034: Spanish (Spain)
036: Hungarian
038: Yugoslavian
039: Italian
041: Switzerland
042: Slovak
044: United Kingdom
045: Danish
046: Swedish
047: Norwegian
048: Polish
049: German
055: Brazil
061: English (International)
351: Portuguese
358: Finnish

Example

```
COUNTRY 001
```

See Also

CODEPAGE

● CODEPAGE

Description

This command defines the code page of international character set.

Syntax

CODEPAGE n

Parameter

n

Description

name or number of code page, which can be divided into 7-bit code page and 8-bit code page further.

7-bit code page name

USA: USA

BRI: British

GER: German

FRE: French

DAN: Danish

ITA: Italian

SPA: Spanish

SWE: Swedish

SWI: Swiss

8-bit code page number

437: United States

850: Multilingual

852: Slavic

860: Portuguese

863: Canadian/French

865: Nordic

Note: Whether to use 7-bit or 8-bit code page is determined by the communication parameter of DATA LENGTH

Example

```
CODEPAGE 437
```

See Also

COUNTRY

● FEED

Description

This command feeds label with the specified length (in dot).

Syntax

FEED n

<u>Parameter</u>	<u>Description</u>
n	unit: dot $1 \leq n \leq 9999$

Example

FEED 40

Note: 200 DPI: 1 mm = 8 dots
 300 DPI: 1 mm = 12 dots

See Also

BACKFEED

● BACKFEED

Description

To back feed label with the specified length (in dot).

Syntax

BACKFEED n

<u>Parameter</u>	<u>Description</u>
n	unit: dot $1 \leq n \leq 9999$

Example

BACKFEED 40

CAUTION: *Impropriety back feed value may cause paper jam or wrinkle.*

Note: 200 DPI: 1 mm = 8 dots
300 DPI: 1 mm = 12 dots

See Also

FEED

● FORMFEED

Description

This command feeds label to the beginning of next label.

Syntax

FORMFEED

<u>Parameter</u>	<u>Description</u>
N/A	N/A

Example

FORMFEED

See Also

FEED

● HOME

Description

It is not expected the first label will be printed on the right position when the printer power is turned on. This command will feed the label to find the gap/black mark first and then back feed to the beginning of label, on condition that the size of the label was set in advance.

Syntax

HOME

<u>Parameter</u>	<u>Description</u>
N/A	N/A

Example

HOME

● PRINT

Description

This command prints the label form stored in the image buffer.

Syntax

PRINT m [,n]

Parameter

m

Description

Specifies how many sets of labels will be printed.

$1 \leq m \leq 999999999$

if m=?, this parameter instructs the portable keyboard to prompt user how many sets and copies of label to print.

Printer will send

~#How many sets?~&<ESC>0x090x0110x0D0x0A

~#How many copies?~&<ESC>0x090x0110x0D0x0A


n

Specifies how many copies should be printed for each set of label.

$1 \leq n \leq 999999999$

Example

```
SIZE 60 mm, 20 mm
SET COUNTER @1 1
@1="0001"
CLS
TEXT 10,10,"3",0,1,1,@1
PRINT 3,2
```



The following example will prompt user "How many sets?" and "How many

copies" on the portable LCD display.

```
DOWNLOAD "DEMO.BAS"  
SIZE 4,1  
GAP 0.12,0  
CLS  
TEXT 100,10,"3",0,1,1,"Portable keyboard Test Print"  
PRINT ?  
EOP
```

See Also

SET COUNTER, INPUT, DOWNLOAD

● CUT

Description

At this command, the printer will activate the cutter to cut the labels immediately without back feeding the label.

Syntax

CUT

<u>Parameter</u>	<u>Description</u>
None	N/A

See Also

SET CUTTER

- **REM**

Description

Comment. Anything beginning with “REM” is ignored by the printer.

Syntax

REM

Example

```
REM *****  
REM This is a demonstration program*  
REM *****  
DOWNLOAD "REMARK.BAS"  
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET CUTTER OFF  
SET PEEL OFF  
CLS  
TEXT 50,50,"3",0,1,1,"REMARK DEMO PROGRAM"  
EOP
```

Label Formatting Commands

- **CLS**

Description

This command clears the image buffer.

Syntax

CLS

<u>Parameter</u>	<u>Description</u>
N/A	N/A

Note: This command must be placed after SIZE command.

Example

CLS

See Also

SIZE

● BAR

Description

This command is used to draw a line or a bar on the label form.

Syntax

BAR x, y, width, height

<u>Parameter</u>	<u>Description</u>
x	The upper left corner x-coordinate in dot
y	The upper left corner y-coordinate in dot
width	The width of bar in dot
height	The height of bar in dot

Note: 200 DPI: 1 mm = 8 dots
300 DPI: 1 mm = 12 dots

Example

BAR 100, 100, 300, 200



See Also

BOX

● ERASE

Description

This command is used to blot out a specified region from image.

Syntax

```
ERASE X_start, Y_start, X_width, Y_height
```

<u>Parameter</u>	<u>Description</u>
X_start	The x-coordinate of the starting point in dot
Y_start	The y-coordinate of the starting point in dot
X_width	The region width in x-axis direction in dot
Y_height	The region height in y-axis direction in dot

Example

```
ERASE 100,100,200,200
```

See Also

CLS

● REVERSE

Description

This command is used to reverse a region of image.

Syntax

```
REVERSE X_start, Y_start, X_width, Y_height
```

<u>Parameter</u>	<u>Description</u>
X_start	The x-coordinate of the starting point in dot
Y_start	The y-coordinate of the starting point in dot
X_width	The region width in x-axis direction in dot
Y_height	The region height in y-axis direction in dot

Example

```
REVERSE 100,100,200,200
```

- **BOX**

Description

This command is used to draw rectangles on the label.


Syntax

BOX X_start, Y_start, X_end, Y_end, line thickness

<u>Parameter</u>	<u>Description</u>
X_start	Specify x-coordinate of upper left corner in dot
Y_start	Specify y-coordinate of upper left corner in dot
X_end	Specify x-coordinate of lower right corner in dot
Y_end	Specify y-coordinate of lower right corner in dot
line thickness	Line thickness of the box in dot

Example

BOX 100,100,200,200,5



See Also

BAR

● BITMAP

Description

This command is used to draw bitmap images (Not BMP graphic file).

Syntax

BITMAP X, Y, width, height, mode, bitmap data...

<u>Parameter</u>	<u>Description</u>
X	Specify the x-coordinate of the bitmap image
Y	Specify the y-coordinate of the bitmap image
width	The width of the image in bytes
height	The height of the image in dot
mode	Graphic mode is listed below:
0	OVERWRITE
1	OR
2	XOR
bitmap data	The bitmap data

Example

```
BITMAP 100,100,10,1,2,1111111111
```

See Also

PUTBMP, PUTPCX

● PUTBMP

Description

This command is used to print BMP format image.

Syntax

```
PUTBMP X, Y, "filename"
```

<u>Parameter</u>	<u>Description</u>
X	The x-coordinate of the BMP format image
Y	The y-coordinate of the BMP format image
filename	The downloaded BMP filename. Case sensitive

Note: *Recommend two-color format (black and white). Maximum 256-colors*

Example

```
PUTBMP 100,100,"LOGO.BMP"
```

● PUTPCX

Description

This command is used to print PCX format image.

Syntax

```
PUTPCX X, Y, "filename"
```

<u>Parameter</u>	<u>Description</u>
X	The x-coordinate of the PCX image
Y	The y-coordinate of the PCX image
filename	The downloaded PCX filename. Case sensitive

Note: *Recommend two-color format (black and white). Maximum 256-colors*

Example

```
PUTPCX 100,100,"LOGO.PCX"
```

See Also

BITMAP, PUTPCX

● **BARCODE**

Description

This command is used to print 1D barcodes on label form.
The available bar codes are listed below:

- Code 128 (switching code subset automatically)
- Code 128M (switching code subset manually)
- EAN 128 (switching code subset automatically)
- Interleaved 2 of 5
- Interleaved 2 of 5 with check digit
- Code 39
- Code 39 with check digit
- Code 93
- EAN 13
- EAN 13 with 2 digits add-on
- EAN 13 with 5 digits add-on
- EAN 8
- EAN 8 with 2 digits add-on
- EAN 8 with 5 digits add-on
- Codabar
- Postnet
- UPC-A
- UPC-A with 2 digits add-on
- UPC-A with 5 digits add-on
- UPC-E
- UPC-E with 2 digits add-on
- UPC-E with 5 digits add-on
- MSI
- PLESSEY
- CPOST
- ITF14
- EAN14
- KIX CODE

Syntax

BARCODE X, Y, "code type", height, human readable, rotation, narrow, wide, "code"

<u>Parameter</u>	<u>Description</u>
X	Specify the x-coordinate of the bar code on label
Y	Specify the y-coordinate of the bar code on label
code type	
128	Code 128, switching code subset A, B, C automatically
128M	Code 128, switching code subset A, B, C manually.

Control code	A	B	C
096	FNC3	FNC3	NONE
097	FNC2	FNC2	NONE
098	SHIFT	SHIFT	NONE
099	CODE C	CODE C	NONE
100	CODE B	FNC4	CODE B
101	FNC4	CODE A	CODE A
102	FNC1	FNC1	FNC1
103	Start (CODE A)		
104	Start (CODE B)		
105	Start (CODE C)		

Use "!" as a starting character for the control code followed by three control codes.

If the start subset is not set, the default starting subset is B.

EAN128	Code 128, switching code subset A, B, C automatically
25	Interleaved 2 of 5
25C	Interleaved 2 of 5 with check digits
39	Code 39
39C	Code 39 with check digits
93	Code 93
EAN13	EAN 13
EAN13+2	EAN 13 with 2 digits add-on
EAN13+5	EAN 13 with 5 digits add-on
EAN8	EAN 8
EAN8+2	EAN 8 with 2 digits add-on
EAN8+5	EAN 8 with 5 digits add-on
CODA	Codabar
POST	Postnet
UPCA	UPC-A
UPCA+2	UPC-A with 2 digits add-on
UPCA+5	UPC-A with 5 digits add-on

UPCE	UPC-E
UPCE+2	UPC-E with 2 digits add-on
UPCE+5	UPC-E with 5 digits add-on
CPOST	China post code
MSI	MSI code
PLESSEY	PLESSEY code
ITF14	ITF 14 code
EAN14	EAN 14 code
KIX	KIX code

height	bar code height in dot
human readable	0: human not readable 1: human readable
rotation	Rotate bar code clockwise in degrees
0	non rotation
90	rotate 90 degrees clockwise
180	rotate 180 degrees clockwise
270	rotate 270 degrees clockwise
narrow	width of narrow element in dot
wide	width of wide element in dot

	Narrow : wide 1:1	Narrow : wide 1:2	narrow : wide 1:3	narrow : wide 2:5
128	10x	-	-	-
EAN128	10x	-	-	-
25	-	10x	10x	5x
25C	-	10x	10x	5x
39	-	10x	10x	5x
39C	-	10x	10x	5x
93	-	-	10x	-
EAN13	8x	-	-	-
EAN13+2	8x	-	-	-
EAN13+5	8x	-	-	-
EAN 8	8x	-	-	-
EAN 8+2	8x	-	-	-
EAN 8+5	8x	-	-	-
CODA	-	10x	10x	5x
POST	1x	-	-	-
UPCA	8x	-	-	-
UPCA+2	8x	-	-	-
UPCA+5	8x	-	-	-
UPCE	8x	-	-	-
UPCE+2	8x	-	-	-
UPCE+5	8x	-	-	-
CPOST	-	10x	10x	5x
MSI	-	-	-	-
PLESSY				

ITF14				
EAN14				
KIX				

code number

the bar code content

Barcode type	Code context Length
128	-
EAN128	-
25	-
25C	-
39	-
39C	-
93	-
EAN13	12
EAN13+2	14
EAN13+5	17
EAN 8	7
EAN 8+2	9
EAN 8+5	12
CODA	-
POST	-
UPCA	
UPCA+2	
UPCA+5	
UPCE	
UPCE+2	
UPCE+5	
CPOST	
MSI	
PLESSY	
ITF14	
EAN14	
KIX	

Example

BARCODE 100,100,"39",96,1,0,2,4,"1000"

BARCODE 10,10,"128M",48,1,0,2,2,"!104!096ABCD!101EFGH"

(The above example of code 128M encoded with CODE B start character. The next character will be the code 128 function character FNC3 which is then followed by the ABCD characters and EFGH characters encoded as CODE A subset.)

● DMATRIX

Description

This command is used to define the DataMatrix 2D bar code.

Syntax

DMATRIX x, y, width, height, [Xm, Lm], expression

<u>Parameter</u>	<u>Description</u>
x	Horizontal start position in dot
y	Vertical start position in dot
width	The width of barcode area in dot
height	The height of barcode area in dot
Xm	Module size in dot
Lm	Expression length (without double quote), $1 \leq m \leq 2048$

Example

```
DMATRIX 10,10,400,400,"DMATRIX"  
DMATRIX 10,10,400,400,L7,DMATRIX
```

● MAXICODE

Description

This command is used to define a 2D Maxicode.

Syntax

MAXICODE x, y, mode, [class, country, post, Lm,] "message"

For mode 2 or 3:

MAXICODE x, y, mode, class, country, postal code, "low priority message"

if country is 840, the postal code is in 99999,9999 format.

For other countries, the code is up to 6 alphanumeric characters.

For mode 4,5,6

MAXICODE x, y, mode, [Lm], "message"

* AIM special format is supported, see page 23 in the spec.

<u>Parameter</u>	<u>Description</u>
x	X-coordinate of the starting point in dot
y	Y-coordinate of the starting point in dot
mode	2,3,4,5
class	Class of service, 3-digit number (for mode 2,3)
country	Country code, 3-digit number (for mode 2,3)
post	Post code (for mode 2,3) Mode 2: (USA) 5-digit+ 4-digit number Mode3: (Canada) 6 alphanumeric post code included by double quotes.
Lm	Expression length (double quote is ignored) , $1 \leq m \leq 138$
message	Barcode content

Example

For USA:

```
MAXICODE 100,100,2,300,840,068107317,"DEMO FOR MAXICODE"
```

For Canada:

```
MAXICODE 100,100,3,300,840,"107317","DEMO FOR MAXICODE"
```

EXAMPLES:

```
REM MODE 4
```

```
SPEED 6
```

```
CLS
```

```
SIZE 4.00,3.00
```

GAP 0.10,0
DENSITY 10
MAXICODE 24,24,4,"THIS IS A 93 CHARACTER CODE SET A MESSAGE
THAT FILLS A MODE 4, UNAPPENDED, MAXICODE SYMBOL..."
BOX 424,16,700,60,2
DIRECTION 0
PRINT 1

REM MODE 5
SPEED 6
CLS
SIZE 4.00,3.00
GAP 0.10,0
DENSITY 10
REM SET MAXIMODE 5
MAXICODE 24,24,5,"THIS IS A 93 CHARACTER CODE SET A MESSAGE
THAT FILLS A MODE 4, UNAPPENDED, MA"
BOX 424,16,700,60,2
DIRECTION 0
PRINT 1

● PDF417

Description

This command is used to define a PDF417 2D barcode.

Syntax

PDF417 x, y, width, height, rotate, [option], expression

<u>Parameter</u>	<u>Description</u>
x	X-coordinate of the starting point in dot
y	Y-coordinate of the starting point in dot
width	The width of barcode in dot
height	The height of barcode in dot
rotate	Rotation counterclockwise.
0:	0 degree
90:	90 degrees
180:	180 degrees
270:	270 degrees
expression label.	Barcode text or string expression to be printed on label.
[option]	
E	Error correction level Range: 0~8
W	Module width in dot Range: 2~9
H	Bar height in dot Range: 4~99
R	Maximum number of rows
C	Maximum number of columns
T	Truncation. 0: Not truncated 1: Truncated
Lm	Expression length (without double quote) , $1 \leq m \leq 2048$

Example

PDF417 100,200,200,300,0,E1,"abcdef"

PDF417 100,200,200,300,0,E1,L6,abcdef

● TEXT

Description

This command is used to print text on label

Syntax

TEXT X, Y, "font", rotation, x-multiplication, y-multiplication, "content"

<u>Parameter</u>	<u>Description</u>
X	The x-coordinate of the text
Y	The y-coordinate of the text
font:	font name
1	8 x 12 fixed pitch dot font
2	12 x 20 fixed pitch dot font
3	16 x 24 fixed pitch dot font
4	24 x 32 fixed pitch dot font
5	32 x 48 dot fixed pitch font
6	14 x 19 dot fixed pitch font OCR-B
7	21 x 27 dot fixed pitch font OCR-B
ROMAN.TTF	ROMAN True type font
rotation:	The rotation angle of text
0	0 degree
90	90 degrees, in clockwise direction
180	180 degrees, in clockwise direction
270	270 degrees, in clockwise direction
x-multiplication:	Horizontal multiplication, up to 10x. Available factors: 1~10 For true type font, this parameter is ignored.
y-multiplication:	Vertical multiplication, up to 10x. Available factors: 1~10 For true type font, this parameter is used to specify the height (point) of true type font. 1 point=1/72 inch.

Example

```
TEXT 100,100,"3",0,1,1,"DEMO FOR TEXT"
```

```
TEXT 100,400,"ROMAN",0,1,1,"True Type Font Test Print"
```


Status Polling Commands (RS-232)

- <ESC>!?

Description

This command is used to obtain the printer status. An inquiry request is solicited by sending an <ESC> (ASCII 27, escape character) as the beginning control character to the printer. It can be sent any time, even in the event of printer error. One byte character is returned, of which one bit is used to flag the printer's current readiness status. If 0 is returned, the printer is ready to print labels.

<u>Bit</u>	<u>Status</u>
0	Head opened
1	Paper jam
2	Out of paper
3	Out of ribbon
4	Pause
5	Printing
6	Cover opened (option)
7	Environment Temperature over range (option)

Syntax

<ESC>!?

<u>Parameter</u>	<u>Description</u>
-	N/A

- **<ESC>!R**

Description

This command is used to reset the printer. It can be sent at any time as long as the printer is powered on and not in the dump mode. The beginning of the command is an ESCAPE character (ASCII 27). The files downloaded in memory will be deleted.

Syntax

<ESC>!R

<u>Parameter</u>	<u>Description</u>
N/A	N/A

- **~!T**

Description

This command is used to inquire the model name and number of the printer. They are returned in ASCII characters.

<u>Printer type</u>	<u>Returned string</u>
K4500	K4500

Syntax

~!T

<u>Parameter</u>	<u>Description</u>
N/A	N/A

- ~!!

Description

The command is used to inquire the code page and country setting of the printer.

The returned information is given in the following format

codepage, country code

ex: 8 bit: 437, 001

7 bit: USA, 001

Regarding the codepages and country codes supported by the printer, please refer to the **CODEPAGE** and **COUNTRY** command respectively.

Syntax

~!!

<u>Parameter</u>	<u>Description</u>
N/A	N/A

See Also

COUNTRY, CODEPAGE

- **~!F**

Description

This command is used to inquire about files resident in the printer memory and fonts installed in the memory module.

The filename is returned in ASCII characters. Each file name ends with 0x0D 0x0A. The ending character is 0x1A.

Syntax

~!F

<u>Parameter</u>	<u>Description</u>
N/A	N/A

See Also

FILES, KILL

Message Translation Protocols

- **~#**

Description

The beginning identifier of the prompt message is sent from the printer to the portable keyboard. The ending identifier is ~&.

Syntax

~#Prompt~&

<u>Parameter</u>	<u>Description</u>
N/A	N/A

Example

~#SELF TEST~&

See Also

INPUT, OUT

● INPUT

Description

This command is used to receive data through serial port. This command is used with portable keyboard.

Syntax

INPUT ["Prompt", number of digits,] counter

<u>Parameter</u>	<u>Description</u>
Prompt	The prompt string is shown on keyboard LCD screen. The maximum length of prompt string is 20 characters.
Number of digits counter	Maximum number of character is 255. The counter to receive input data.

Note: Prior to use the counter as a variable, please declare the counter first by SET COUNTER command.

Example

(1). With prompt

```
SET COUNTER @1 0
@1="123"
INPUT "HELLO",@1
```

The printer will send ~#HELLO~&1230x0D0x0A through RS-232 to instruct portable keyboard to display the prompt "HELLO" on the first line of LCD display and cursor will be placed at the second line of LCD display to wait for input data.

(2). Without prompt

```
SET COUNTER @1 0
@1="123"
INPUT @1
```

Printer will send ~#~&1230x0D0x0A through RS-232 to portable keyboard. No prompt will be displayed on the first line of portable keyboard LCD display, but cursor will be placed at the second line of LCD display waiting for input data.

(3). With prompt, limited characters for input data

```
SET COUNTER @1 0
@1="123"
INPUT "HELLO",5,@1
```

Printer will send ~#HELLO~&<ESC>0x051230x0D0x0A through RS-232 to portable LCD display and limit the allowable input data length for 5 characters.

(4). Without prompt, limited characters for input

```
SET COUNTER @1 0
@1="123"
INPUT 5,@1
```

Printer will send ~#~&<ESC>0x051230x0D0x0A

No prompt will be displayed on the first line of portable keyboard LCD display, but cursor will be placed at the second line of LCD display waiting for input data. The maximum numbers of characters to input is limited to 5 characters.

(5). Complete example

```
DOWNLOAD "DEMO.BAS"
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL OFF
SET COUNTER @1 0
@1=""
CLS
INPUT "PLEASE ENTER CODE",@1
BARCODE 20,100,"39",48,1,0,2,5,@1
PRINT ?
EOP
```


● OUT

Description

This command is used to send data through printer serial port. 0x0D and 0x0A will be attached at the end of prompt string.

Syntax

OUT "prompt", counter

<u>Parameter</u>	<u>Description</u>
prompt	Prompt string send out through printer RS-232.
counter	The output message

Example

```
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET COUNTER @1 0  
@1="1234"  
CLS  
OUT "PRICE:" ,@1
```

- **BEEP**

Description

This command is used to issue a beep sound on portable keyboard. Printer sends 0x07 to portable keyboard.

Syntax

BEEP

<u>Parameter</u>	<u>Description</u>
N/A	N/A

Example

```
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET CUTTER OFF  
SET PEEL OFF  
CLS  
BEEP
```

Commands for Windows Driver

- **!B**

Description

This command is used to store bitmap image data in the memory. Behind the nnn is the bitmap data.

Syntax

!Bnnn

<u>Parameter</u>	<u>Description</u>
nnn	The number of bytes of image data sent from PC to printer, expressed in 3 decimal digits.

Example

!B100

See Also

BITMAP

- **!J**

Description

This command is used to print the bitmap data at the specified position (in y-direction).

Syntax

!Jnnnn

Parameter

nnnn

Description

Print image at the specified position in y-direction.
The position is expressed in 4 decimal digits.

Example

!J0100

See Also

FEED

File Management Commands

● DOWNLOAD

Description

“DOWNLOAD” is a header of the file that is to be saved in the printer's memory.

The downloaded files can be divided to two categories: program file and data file (including text data file, PCX graphic files and bitmap font file)

The detailed descriptions regarding the download syntax for different files are shown below:

Maximum numbers of file saved in DRAM: 50

Maximum numbers of file saved in Flash Files: 100

Syntax

1. Download a program file

DOWNLOAD [F,]“FILENAME.BAS”

F: Write to flash.

If this parameter is been ignored, data file will be downloaded to DRAM.

FILENAME.BAS The filename resident in printer memory.

Note: (1). The filename is case sensitive.

(2). The extension of of the program file must be “.BAS”

(3). Filename format must be in 8.3 format.

2. Download a data file

DOWNLOAD [F,]“FILENAME”, DATA SIZE, DATA CONTENT...where

F: Write to flash

if this parameter is been ignored, data file will be downloaded to DRAM.

FILENAME: The name of data file that will remain resident in the printer memory. It is case sensitive.

DATA SIZE: The actual size (numbers of byte) of the data file without header.

Note: For text data file, CR (carriage return) 0x0D and LF (Line Feed) 0x0A is the separator of data.

Example

The program listed below will download to printer SDRAM.

```
DOWNLOAD "EXAMPLE.BAS"  
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET CUTTER OFF  
SET PEEL OFF  
CLS  
TEXT 100,100,"3",0,1,1,"EXAMPLE PROGRAM"  
PRINT 1  
EOP
```

Note: *When writing a download program, "DOWNLOAD" header must be placed in the beginning of file, and "EOP" must be placed at the end of program.*

Below is an example of downloading data file.

```
DOWNLOAD "DATA" , 20 , COMPUTER<ENTER>  
2001<ENTER>  
21<ENTER>
```

Note: *<ENTER> stands for keyboard "ENTER" key. In the above example, please press "ENTER" key instead of typing <ENTER> in the above example.*

See Also

EOP, RUN

- **EOP**

Description

End of program. To declare the start and end of BASIC language commands used in a program, the “DOWNLOAD “FILENAME.BAS” must be added in the first line of the program, and “EOP” statement at the last line of program.

Syntax

EOP

Example

```
DOWNLOAD "DEMO.BAS"  
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET CUTTER OFF  
SET PEEL OFF  
CLS  
TEXT 100,100,"3",0,1,1,"DEMO PROGRAM"  
PRINT 1  
EOP
```

See Also

DOWNLOAD

● FILES

Description

This command lists the files that remained in the printer memory (both FLASH memory and DRAM). The total memory size and available memory size are listed as well.

Syntax

FILES

Example

Follow the steps below to list the files that are saved in printer memory in DOS environment through serial port connection

```
.  
C:\>MODE COM2 96,N,8,1<ENTER>  
C:\>COPY CON COM2<ENTER>  
FILES<ENTER>  
^Z<ENTER>  
C:\>
```

Note: <ENTER> stands for keyboard “ENTER” key. In the above example, please press “ENTER” key instead of typing <ENTER> in the above example.

See Also

~!F, KILL

- **KILL**

Description

This command deletes a file in the printer memory. The wild card (*) will delete all files resident in both DRAM and FLASH memory.

Syntax

```
KILL "FILENAME"  
KILL "*.PCX"  
KILL "**"
```

Example

Users can use printer SELF TEST utility to list printer configurations and files saved in the printer memory, or use the FILES command to inquire the files saved in printer.

Follow the steps below to delete files in the printer memory via serial port connection

```
.  
C:\>MODE COM2 96,N,8,1<ENTER>  
C:\>COPY CON COM2<ENTER>  
FILES<ENTER>  
C:\>COPY CON COM2<ENTER>  
    KILL "DEMO.BAS" <ENTER>  
    ^Z<ENTER>  
C:\>FILES<ENTER>
```

Note: <ENTER> stands for keyboard "ENTER" key. In the above example, please press "ENTER" key instead of typing <ENTER> in the above example.

See Also

~!F, FILES

- **RUN**

Description

This command is used to execute a program that resident in printer memory

Syntax

RUN "FILENAME.BAS"

Example

```
C:\>COPY CON LPT1<ENTER>  
RUN "DEMO.BAS"<ENTER>  
^Z<ENTER>  
C:\>
```

Note: <ENTER> stands for keyboard "ENTER" key. In the above example, please press "ENTER" key instead of typing <ENTER> in the above example.

See Also

DOWNLOAD, EOP

Device Reconfiguration Commands

● SET COUNTER

Description

Counter can be a real counter or a variable.
This setting sets the counter number in program and their increments.
There are three different kind of counters: digit (0~9~0), lower case letter (a~z~a) and upper case letter (A~Z~A).

Syntax

```
SET COUNTER @n step  
@n = "Expression"
```

<u>Parameter</u>	<u>Description</u>
@n	n: counter number. There are 50 counters available (0~49) in the printer.
step	The increment of the counter, can be positive or negative. -999999999<= step <=999999999 If the counter is used as a fixed variable, please set the increment to 0.
Expression	Initial string. String length is 2K bytes

Example

```
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
COUNTER @1 1  
@1="00001"  
COUNTER @2 5  
@2="AB000001"  
CLS  
TEXT 50,50,"3",0,1,1,@1  
BARCODE 50,500,"39",48,1,0,2,4,@2  
PRINT 100
```

● SET CUTTER

Description

This setting is used to activate/deactivate the cutter and define how many printed labels to be cut at one time.

Syntax

SET CUTTER OFF/BATCH/pieces

<u>Parameter</u>	<u>Description</u>
OFF	Disable cutter function.
BATCH	Set printer to cut label at the end of printing job.
Pieces	Set number of printing labels per cut. 0<= pieces <=65535

Example

```
REM SET CUTTER FUNCTION OFF EXAMPLE PROGRAM
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL OFF
CLS
TEXT 50,50,"3",0,1,1,"SET CUTTER 0"
BARCODE 50,500,"39",48,1,0,2,4,"DEMO14"
PRINT 1
```

```
REM SET CUTTER BATCH EXAMPLE PROGRAM
REM This program cuts 3 times (3 sets)
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER BATCH
SET PEEL OFF
CLS
TEXT 50,50,"3",0,1,1,"SET CUTTER BATCH"
BARCODE 50,500,"39",48,1,0,2,4,"DEMO14"
PRINT 3,2
EOP
```

```
REM SET CUTTER PIECE EXAMPLE PROGRAM
REM This program cuts each printed label
SIZE 4,4
GAP 0,0
```

DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER 1
SET PEEL OFF
CLS
TEXT 50,50,"3",0,1,1,"SET CUTTER PIECE"
BARCODE 50,500,"39",48,1,0,2,4,"DEMO16"
PRINT 3,2

● SET PEEL

Description

This setting is used to enable/disable the self-peeling function.

The default setting for this function is off. When this function is set on, the printer stops after each label printing, and does not print the next label until the peeled label is taken away.

Syntax

SET PEEL ON/OFF

<u>Parameter</u>	<u>Description</u>
ON	Enable the self-peeling function
OFF	Disable the self-peeling function

Example

```
REM SELF-PEELING FUNCTION ON
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL ON
CLS
TEXT 50,100,"3",0,1,1,"SELF-PEELING FUNCTION TEST"
PRINT 5
```

See Also

OFFEST

● SET TEAR

Description

This command is used to enable/disable feeding label to gap position for tearing off.

Syntax

SET TEAR ON/OFF

Parameter

ON

OFF

Description

The label gap will stop at the tear off position after print.

The label gap will NOT stop at the tear off position after print. The beginning of label will be aligned to print head.

Example

```
REM SELF-PEELING FUNCTION ON
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET TEAR ON
CLS
TEXT 50,100,"3",0,1,1,"SELF-PEELING FUNCTION TEST"
PRINT 5
```

● SET GAP

Description

This setting is used to set the gap sensor light emission sensitiv. The printer initiates automatic gap sensor calibration as you hold down the PAUSE key and then turn on the printer power. But this function may cease to work if the thickness of the backing paper and that of label with backing paper are not of appreciable difference to the sensor, or when there are pre-printed marks or patterns on the label. In such case, users have to calibrate the gap sensor manually by this command. This is a trial-and-error method to attain the proper setting.

Syntax

SET GAP n

<u>Parameter</u>	<u>Description</u>
n	The gap sensor light emission strength. Available range are from 0 to 255

Example

The example below is operated in DOS environment via the serial port connection.

```
C:\>MODE COM2 96,N,8,1<ENTER>
```

```
C:\>COPY CON COM2<ENTER>
```

```
    SET GAP 1<ENTER>
```

```
    ^Z<ENTER>
```

```
C:\>
```

Note: <ENTER> stands for keyboard "ENTER" key. In the above example, please press "ENTER" key instead of typing <ENTER> in the above example.

Press the FEED key to test. Does printer stop at the same position on each label without the error light blinking? If not, please adjust the setting to a larger number again.

When trying with this setting, please begin from 0 and then on to higher values gradually.

● SET RIBBON

Description

This setting is used to enable/disable ribbon sensor detection. (Thermal Transfer Printing/Thermal Direct Printing)

Syntax

SET RIBBON ON /OFF

<u>Parameter</u>	<u>Description</u>
ON	Thermal transfer printing
OFF	Thermal direct printing

Example

```
REM Thermal direct printing
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL OFF
SET RIBBON OFF
CLS
BARCODE 100,100,"39",48,1,0,2,5,"CODE 39"
PRINT 1
```

● SET COM1

Description

This setting defines communication parameters for printer serial port.

Syntax

SET COM1 baud, parity, data, stop

<u>Parameter</u>	<u>Description</u>
baud	Baud rate, available baud rates are as listed : 24: 2400 bps 48: 4800 bps 96: 9600 bps 19: 19200 bps 38: 38400 bps 56: 56000 bps
parity	Parity check N: None parity check E: Even parity check O: Odd parity check
data	Data bit 8: 8 bits data 7: 7 bits data
stop	Stop bit 1: 1 stop bit 2: 2 stop bits

Example

The parallel port is used to setup the printer serial port in this example by MS-DOS mode.

```
C:\>COPY CON LPT1<ENTER>  
SET COM1 19,N,8,1<ENTER>  
^Z<ENTER>  
C:\>
```

Note: <ENTER> stands for keyboard "ENTER" key. In the above example, please press "ENTER" key instead of typing <ENTER> in the above example.

Printer Global Variables

- **@LABEL**

Description

This variable (read only) is used to count how many pieces of labels have been printed. It can't be initialized if the printer is reset. It will be memorized if the printer power is turned off.

Syntax

@LABEL

0<=@LABEL<=999999999

Example

```
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL OFF
SET RIBBON OFF
CLS
TEXT 100 ,100 , "3" ,0,1,1,@LABEL
PRINT 1
```

- **@YEAR**

Description

This variable is used to read from/write to RTC the year data. Two-digit (00~99) year format is supported by RTC as well as the leap year and year 2000.

Syntax

Write attribute: @YEAR="01"

Read attribute: @YEAR

Range: 00~99

Example

```
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL OFF
CLS
TEXT 10,10,"3",0,1,1,@YEAR
PRINT 1
```

See Also

@MONTH, @DATE, @DAY, @HOUR, @MINUTE, @SECOND

- **@MONTH**

Description

This variable is used to read from/write to RTC the month data. Two-digits (01~12) month format is supported by RTC.

Syntax

Write attribute: @MONTH="01"

Read attribute: @MONTH

Range: 01~12

Example

```
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET CUTTER OFF  
SET PEEL OFF  
CLS  
TEXT 10,10,"3",0,1,1,@MONTH  
PRINT 1
```

See Also

@YEAR, @DATE, @DAY, @HOUR, @MINUTE, @SECOND

- **@DATE**

Description

This variable is used to read from/write to RTC the date data. Two-digits (01~31) date format is supported by RTC.

Syntax

Write attribute: @DATE="12"

Read attribute: @DATE

Range: 01~31

Example

```
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL OFF
CLS
TEXT 10,10,"3",0,1,1,@DATE
PRINT 1
```

See Also

@YEAR, @MONTH, @DAY, @HOUR, @MINUTE, @SECOND

- **@DAY**

Description

This variable is used to read from/write to RTC the week data, which is represented by one single digit (1~7).. .

Syntax

Write attribute: @DAY="3"

Read attribute: @DAY

Range:

0(Sunday)~6(Saturday)

Example

```
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL OFF
CLS
TEXT 10,10,"3",0,1,1,@DAY
PRINT 1
```

See Also

@YEAR, @MONTH, @DATE, @HOUR, @MINUTE, @SECOND

- **@HOUR**

Description

This variable is used to read from/write to RTC the hour data. The 24-hour-day system (00~23) is supported by RTC.

Syntax

Write attribute: @HOUR ="12"

Read attribute: @HOUR

Range: 00~23

Example

```
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL OFF
CLS
TEXT 10,10,"3",0,1,1,@HOUR
PRINT 1
```

See Also

@YEAR, @MONTH, @DATE, @DAY, @MINUTE, @SECOND

- **@MINUTE**

Description

This variable is used to read from/write to RTC the minute data. Two-digits (00~59) minute format is supported by RTC.

Syntax

Write attribute: @MINUTE ="12"

Read attribute: @MINUTE

Range: 00~59

Example

```
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET CUTTER OFF  
SET PEEL OFF  
CLS  
TEXT 10,10,"3",0,1,1,@MINUTE  
PRINT 1
```

See Also

@YEAR, @MONTH, @DATE, @DAY, @HOUR, @SECOND

- **@SECOND**

Description

This variable is used to read from/write to RTC the second data. Two-digits (00~59) second format is supported by RTC.

Syntax

Write attribute: @SECOND="12"

Read attribute: @SECOND

Range: 00~59

Example

```
SIZE 4,4
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET CUTTER OFF
SET PEEL OFF
CLS
TEXT 10,10,"3",0,1,1,@SECOND
PRINT 1
```

See Also

@YEAR, @MONTH, @DATE, @DAY, @HOUR, @MINUTE

